http://www.mapleprimes.com/questions/148272-How-To-Find-The-Double-Integral-With-Maple

```
> restart; interface(version);
```

Classic Worksheet Interface, Maple 17.00, Windows, Feb 21 2013, Build ID 813473

```
> (cos(y)*cos(x)+cos(x)+cos(y))/(3+2*cos(x)+2*cos(y)+2*cos(y)*cos(x))^(3/2):
  L:=unapply(%, x,y);
  ``;
  '1/Pi/4*Int(Int(L(x,y),y = 0 .. 2*Pi),x = 0 .. 2*Pi)': '%'=evalf(%);
```

$$L := (x, y) \to \frac{\cos(y)\cos(x) + \cos(x) + \cos(y)}{(3 + 2\cos(x) + 2\cos(y) + 2\cos(y)\cos(x))^{(3/2)}}$$

$$\frac{1}{4}\left(\frac{1}{\pi}\int_0^{2\pi}\int_0^{2\pi} L(x, y)\, dy\, dx\right) = -0.999999999999995$$

```
> myRange:= 0 .. 2*Pi;
  #myRange:= 0 .. Pi;
  plot3d(L(x,y), y=myRange, x=myRange, axes=boxed, orientation=[-135,30]):
  plots:-spacecurve([t,t, 1/9], t=myRange, axes=boxed, color=red, thickness=3):
  plots:-display(%%,%);

  'L(0,0)':  '%'=simplify(%);
  '[L(Pi-x,y)=L(Pi+x,y),L(x,Pi-y)=L(x,Pi+y), L(x,y)=L(y,x)]'; map(is,%);
```

myRange := 0 .. 2 π



$$L(0, 0) = \frac{1}{9}$$

$$[L(\pi - x, y) = L(\pi + x, y), L(x, \pi - y) = L(x, \pi + y), L(x, y) = L(y, x)]$$

[true, true, true]

By symmetry reduce to the first quadrant

```
> '1/Pi*Int(Int(L(x,y),y = 0 .. Pi),x = 0 .. Pi)': '%'=evalf(%);
```

$$\frac{1}{\pi}\int_0^{\pi}\int_0^{\pi} L(x, y)\, dy\, dx = -0.999999999999994$$

```
> `in`(x,RealRange(0,Pi)), `so use base change`, cos(x)+1 = (2*t)^2/2;
```

$x \in$ RealRange(0, π), so use base change, $\cos(x) + 1 = 2\,t^2$

```
> 'Int(L(x,y),y = 0 .. Pi)': 'Int(%,x = 0 .. Pi)';
  ``=%;
```

```
value(%): subs(int=Int, %);
rhs(%):
Change(%, cos(x)+1 = (2*t)^2/2, t): subs(t=x, %): combine(%):
``=map(collect,%, EllipticK);
convert(%, hypergeom);
V:=rhs(%):

``=evalf(V);
```

$$\int_0^\pi \int_0^\pi L(x, y)\, dy\, dx$$

$$= \int_0^\pi \int_0^\pi \frac{\cos(y)\cos(x) + \cos(x) + \cos(y)}{(3 + 2\cos(x) + 2\cos(y) + 2\cos(y)\cos(x))^{(3/2)}}\, dy\, dx$$

$$= \int_0^\pi \frac{4\,\mathrm{EllipticK}(2\sqrt{-\cos(x) - 1})\cos(x) + 5\,\mathrm{EllipticK}(2\sqrt{-\cos(x) - 1}) - 3\,\mathrm{EllipticE}(2\sqrt{-\cos(x) - 1})}{4\cos(x) + 5}\, dx$$

$$= \int_0^1 \frac{2\,\mathrm{EllipticK}(2\,I\,x\sqrt{2})}{\sqrt{-x^2 + 1}} - \frac{6\,\mathrm{EllipticE}(2\,I\,x\sqrt{2})}{\sqrt{-x^2 + 1}\,(8\,x^2 + 1)}\, dx$$

$$= \int_0^1 \frac{\pi\,\mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{1}{2}\right], [1], -8\,x^2\right)}{\sqrt{-x^2 + 1}} - \frac{3\,\pi\,\mathrm{hypergeom}\left(\left[\frac{-1}{2}, \frac{1}{2}\right], [1], -8\,x^2\right)}{\sqrt{-x^2 + 1}\,(8\,x^2 + 1)}\, dx$$

$$= -3.14159265358976$$

```
> op(1, V): A,B:=op(%);
``;``;
B1:='convert(B, hypergeom, "raise a")'; #convert(%, hypergeom, "lower b");
``=%;
```

$$A, B := \frac{\pi\,\mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{1}{2}\right], [1], -8\,x^2\right)}{\sqrt{-x^2 + 1}}, -\frac{3\,\pi\,\mathrm{hypergeom}\left(\left[\frac{-1}{2}, \frac{1}{2}\right], [1], -8\,x^2\right)}{\sqrt{-x^2 + 1}\,(8\,x^2 + 1)}$$

$$B1 := \mathrm{convert}(B, \mathrm{hypergeom}, \text{"raise a"})$$

$$= -\frac{3\,\pi\,\mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{3}{2}\right], [1], -8\,x^2\right)}{\sqrt{-x^2 + 1}}$$

```
> 'Int(Int(L(x,y),y = 0 .. Pi),x = 0 .. Pi)';
``='Int(A+B, x=0 .. 1)';
``='Int(A+B1, x=0 .. 1)';
expand(%): simplify(%, size);
evalf(%);
```

$$\int_0^\pi \int_0^\pi L(x, y)\, dy\, dx$$

$$= \int_0^1 A + B\, dx$$

$$= \int_0^1 A + B1\, dx$$

$$= -\pi \int_0^1 -\frac{\mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{1}{2}\right], [1], -8\,x^2\right) - 3\,\mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{3}{2}\right], [1], -8\,x^2\right)}{\sqrt{-x^2 + 1}}\, dx$$

$$= -3.14159265358979$$

```
> 'Int(A/Pi+B1/Pi, x=0..1)';
```

```
``=%;
value(%);
evalf(%);
```

$$\int_0^1 \frac{A}{\pi} + \frac{B1}{\pi}\, dx$$

$$= \int_0^1 \frac{\mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{1}{2}\right], [1], -8x^2\right)}{\sqrt{-x^2+1}} - \frac{3\,\mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{3}{2}\right], [1], -8x^2\right)}{\sqrt{-x^2+1}}\, dx$$

$$= \frac{1}{8} \frac{\sqrt{2}\left(\mathrm{MeijerG}\left(\left[\left[\frac{1}{2}\right], \left[\frac{1}{2}, \frac{1}{2}\right]\right], [[0, 0, 0], [\ ]], \frac{1}{8}\right) - 6\,\mathrm{MeijerG}\left(\left[\left[\frac{1}{2}\right], \left[\frac{1}{2}, \frac{1}{2}\right]\right], [[1, 0, 0], [\ ]], \frac{1}{8}\right)\right)}{\sqrt{\pi}}$$

$$= -1.00000000000000$$